




## Ejercicio 1

[ 3.5 puntos ]

El juego del **Buscaminas** consiste en despejar todas las casillas de un tablero que no oculten una mina. El tablero inicialmente se muestra sin ninguna información, y el usuario puede ir seleccionando casillas. Si selecciona una con una mina oculta, pierde. Si selecciona una sin mina, se le muestra en la casilla un número que indica cuántas casillas de las colindantes contienen una mina. Así, si una casilla tiene el número 3, significa que de las ocho casillas que hay alrededor (si no está en una esquina o borde, en cuyo caso tendría 3 o 5 casillas colindantes en vez de 8) hay 3 con mina y 5 sin mina. Si se descubre una casilla sin número (con un cero) indica que ninguna de las casillas vecinas tiene mina y dichas casillas vecinas se descubren automáticamente.



Se pide, en lenguaje Pascal:

- Definir una estructura de datos **tpTablero** que permita almacenar el tablero de la partida:
  - Tiene un tamaño máximo de 100x100, pero se puede jugar a tamaños más pequeños.
  - Para cada casilla es necesario almacenar información que indique: si contiene una mina, si el usuario ha descubierto ya dicha casilla o no, y cuántas casillas colindantes contienen una mina.

A partir de la estructura de datos anterior, implementa los siguientes procedimientos y funciones:

- Un procedimiento que genere el tablero inicial de forma aleatoria, con un número de minas especificado por el usuario:

```
procedure genera_tablero(??? tablero: tpTablero; ??? nminas: integer);
```

El procedimiento debe generar un tablero con toda la información necesaria, incluyendo, para cada casilla, si contiene mina o no, si está descubierta o no, y el número de casillas colindantes que contienen una mina.

Puedes usar la función de Pascal estándar (ya existente, no tienes que implementarla):

```
function random(maxval: longint): longint;
```

para generar las posiciones de las minas de forma aleatoria. Esta función toma como parámetro un entero (**longint**) y devuelve un entero aleatorio (**longint**) que está en el rango [0..maxval).

- Un procedimiento que modifique el tablero cuando el jugador descubre una casilla:

```
procedure descubre_casilla(??? tablero: tpTablero; ??? fil,col: integer);
```

El procedimiento debe:

- Asegurarse de que la casilla es válida (está dentro del tablero).
- Marcar como descubierta la casilla correspondiente.
- Si no contiene ninguna casilla vecina con mina, marcar como descubiertas también las casillas colindantes.

Puedes definir estructuras de datos o procedimientos y funciones adicionales si lo consideras necesario.

**Propuesta de solución:**

```
1  program paridad;
2  uses crt;
3
4  const
5      MAX_SIZE = 100;
6  type
7      tpCasilla = record
8          haymina :      boolean;
9          minasvecinas : integer;
10         descubierta :  boolean;
11     end;
12     tpTablero = record
13         n,m : integer;
14         casillas : array[1..MAX_SIZE,1..MAX_SIZE] of tpCasilla;
15     end;
16
17 procedure dibuja_tablero(const t: tpTablero);
18 var
19     i,j : integer;
20 begin
21     clrscr();
22     for i:=1 to t.n do
23     begin
24         write(i:2, ' ');
25         for j:=1 to t.m do
26         begin
27             if t.casillas[i,j].descubierta
28             then begin
29                 if t.casillas[i,j].haymina
30                 then begin
31                     write('*');
32                 end
33                 else begin
34                     write(t.casillas[i,j].minasvecinas);
35                 end;
36             end
37             else begin
38                 write(' ');
39             end;
40         end;
41         writeln();
42     end;
43 end;
44
45 procedure genera_tablero(var t: tpTablero; nminas : integer);
46 var
47     mina: integer;
48     imina,jmina,i,j: integer;
49 begin
50     for i:=1 to t.n do
51         for j:=1 to t.m do
```

```
52     begin
53         t.casillas[i,j].haymina := false;
54         t.casillas[i,j].minasvecinas := 0;
55         t.casillas[i,j].descubierta := false;
56     end;
57
58     for mina:=1 to nminas do
59     begin
60         repeat
61             imina := 1 + random(t.n);
62             jmina := 1 + random(t.m);
63         until not t.casillas[imina,jmina].haymina;
64         t.casillas[imina,jmina].haymina := true;
65         for i:=imina-1 to imina+1 do
66             for j:=jmina-1 to jmina+1 do
67                 begin
68                     if (i>=1) and (i<=t.n)
69                        and (j>=1) and (j<=t.m)
70                        and ((i<>imina) or (j<>jmina))
71                     then
72                         inc(t.casillas[i,j].minasvecinas);
73                 end;
74             end;
75         end;
76
77     procedure descubre_casilla(var t: tpTablero; f,c : integer);
78     var
79         i,j: integer;
80     begin
81         if (f>=1) and (f<=t.n) and (c>=1) and (c<=t.m)
82         then begin
83             t.casillas[f,c].descubierta := true;
84             if t.casillas[f,c].minasvecinas = 0
85             then begin
86                 for i:=f-1 to f+1 do
87                     for j:=c-1 to c+1 do
88                         begin
89                             if (i>=1) and (i<=t.n)
90                                and (j>=1) and (j<=t.m)
91                                and ((i<>f) or (j<>c))
92                             then
93                                 t.casillas[i,j].descubierta := true;
94                         end;
95                     end;
96                 end;
97             end;
98
99     var
100         tablero: tpTablero;
101         f,c: integer;
102     begin
103         randomize();
104         tablero.n := 10;
```

```
105  tablero.m := 20;
106  genera_tablero(tablero,10);
107  dibuja_tablero(tablero);
108  repeat
109      readln(f,c);
110      descubre_casilla(tablero,f,c);
111      dibuja_tablero(tablero);
112  until false;
113  end.
```

## Ejercicio 2

**[ 3.5 puntos ]**

La empresa de transporte público de una ciudad quiere evaluar el uso de sus distintas líneas de autobús por parte de niños y adultos. Para ello, pide a los conductores que apunten, a lo largo de su jornada, para cada persona que sube al autobús, si es un niño o un adulto. La información correspondiente a un día acaba almacenada en un *fichero de texto*, en el que cada línea del fichero contiene información para una ruta de autobús: un entero correspondiente al número de ruta de autobús, un espacio en blanco, y a partir de ahí una secuencia de 'n' y 'a', en la que hay una 'n' por cada niño que sube, y una 'a' por cada adulto que sube. La Figura 1 muestra un ejemplo de un posible fichero de entrada.

Se pide desarrollar un programa Pascal que, dado un *fichero de texto* de entrada '**buses.txt**' con la estructura descrita y que contiene la información correspondiente a un día, genere un *fichero de texto* de salida para ese día '**buses\_procesado.txt**' con la siguiente estructura: cada línea del fichero de salida corresponderá a una ruta de autobús, y tendrá escrito, por este orden, el número de ruta, y luego el porcentaje de niños y porcentaje de adultos que han viajado en esa ruta durante ese día. La Figura 2 muestra un ejemplo del fichero de salida, con el formato que tiene que tener.

Importante: En el fichero de salida, las líneas de autobús deben estar en orden, mientras que en el fichero de entrada no tienen por qué estarlo. Además, el fichero de entrada puede tener varias líneas correspondientes a la misma ruta (distintos conductores que hacen la ruta a lo largo del día).

Observaciones:

- Una ruta puede no haber llevado ningún pasajero. En ese caso, la línea del fichero contendrá únicamente el número de ruta, seguido de un espacio en blanco (por ejemplo, la ruta número 34 en la Figura 1).
- La ciudad cuenta con 26 líneas de autobús, numeradas del 20 al 45.
- El número de veces que cada ruta se hace al día es variable, pero no es relevante, simplemente interesa el total diario (ver la ruta número 23 en el ejemplo).

Figura 1: Ejemplo de fichero de entrada **buses.txt**

```
45 annnaaaana
23 an
20 aann
23 nnanna
34
36 aaaa
```

Figura 2: Ejemplo de fichero de salida **buses\_procesado.txt**

```
20 - n = 50.0%, a = 50.0%
21 -
22 -
23 - n = 62.5%, a = 37.5%
24 -
25 -
...
36 - n = 0.0%, a = 100.0%
...
44 -
45 - n = 40.0%, a = 60.0%
```

**Propuesta de solución:**

```
1  program buses;
2
3  const
4      N = 26; { numero de rutas }
5      RI = 20; { ruta inicial }
6
7  type
8      tpRutas = array[RI..(RI+N-1)] of integer;
9
10 var
11     tIn, tOut: text;
12     vecN, vecA: tpRutas;
13     porN, porA: real;
14     i, ruta: integer;
15     c: char;
16
17 begin
18
19     { Inicializacion de contadores }
20     for i:=RI to (RI+N-1) do begin
21         vecN[i] := 0;
22         vecA[i] := 0;
23     end;
24
25     { Lectura de fichero de entrada y carga de info en los vectores }
26     assign(tIn, 'buses.txt');
27     reset(tIn);
28
29     while not eof(tIn) do begin
30         read(tIn, ruta);
31         read(tIn, c); { lectura del espacio tras el numero de ruta }
32         while not eoln(tIn) do begin
33             read(tIn, c);
34             if (c='n') then
35                 vecN[ruta] := vecN[ruta] + 1
36             else if (c='a') then
37                 vecA[ruta] := vecA[ruta] + 1;
38         end;
39         readln(tIn);
40     end;
41
42     close(tIn);
43
44     { Escritura en el fichero de salida }
45     assign(tOut, 'buses_procesado.txt');
46     rewrite(tOut);
47
48     for i:=RI to (RI+N-1) do begin
49         if (vecN[i]+vecA[i])>0 then begin
50             porN := (vecN[i]/(vecN[i]+vecA[i]))*100;
51             porA := (vecA[i]/(vecN[i]+vecA[i]))*100;
52             writeln(tOut, i, ' - n = ', porN:1:1, '%, a = ', porA:1:1, '%');
53         end
54         else
55             writeln(tOut, i, ' - ');
56     end;
57
58     close(tOut);
```

59

60 **end.**

## Ejercicio 3

[ 3.0 puntos ]

La memoria RAM de los ordenadores no es perfecta, y con el paso del tiempo puede estropearse y corromper la información que guarda (lo que recuperas no es lo mismo que lo que has almacenado).

Esos problemas no se pueden solucionar de forma ideal, pero al menos los computadores incorporan un sistema de comprobación que sirve para averiguar si en el dato almacenado se ha corrompido UN bit. Para ello se calcula la **paridad binaria** del dato almacenado, es decir, si al expresarlo en binario contiene un número PAR de unos (paridad 0) o IMPAR (paridad 1). Así, con un único bit extra, podemos saber si el número ha cambiado respecto al original:

Número (n)	Número en binario (8 bits)	Cantidad de 1s	Paridad
0	00000000	0	0
1	00000001	1	1
5	00000101	2	0
7	00000111	3	1
105	01101001	4	0
211	11010011	5	1

Implementa en Pascal una función:

```
function paridad(n: integer): integer;
```

que calcule la paridad binaria del número que se le pasa como parámetro, es decir, devuelve 0 si al expresarlo en binario el dato contiene un número par de unos, o bien 1 en caso contrario.

Para poder implementar dicha función, hay que saber cómo convertir un número decimal (base 10) a binario (base 2). Para hacer la conversión, se puede: dividir el número decimal entre dos y anotar el resto de dicha división, y hacer esto de forma iterativa hasta que la división entera entre dos da cero. Los valores del resto que hemos ido anotando dan el número en binario, ordenados de forma que el primer resto obtenido es la última cifra del número en binario (y el último resto obtenido es la primera cifra del número en binario). La Figura 1 ilustra el proceso de conversión aquí descrito para dos números de ejemplo.




Figura 1: Conversión de decimal a binario para dos números de ejemplo. En ambos casos, el número en binario se representa con 8 bits.



**Propuesta de solución:**

```
1  program paridad;
2
3  function paridad(n: integer): integer;
4  var
5      contUnos: integer;
6  begin
7      { Conversion de n a binario y vamos contando unos }
8      contUnos := 0;
9      while (n > 0) do begin
10         if (n mod 2 = 1) then contUnos := contUnos + 1;
11         n := n div 2;
12     end;
13
14     if (contUnos mod 2 = 0) then
15         paridad := 0
16     else
17         paridad := 1;
18 end;
19
20 begin
21     writeln(paridad(0));
22     writeln(paridad(1));
23     writeln(paridad(5));
24     writeln(paridad(7));
25     writeln(paridad(105));
26     writeln(paridad(211));
27 end.
```